
Subject: Issues with recalculating wealth index
Posted by cw21 on Fri, 28 May 2021 17:27:36 GMT
View Forum Message <> Reply to Message

I am currently trying to reconstruct the wealth index for multiple countries (see excel attachment
for specific countries) removing the cooking fuel variables, following the steps to constructing the
new DHS Wealth Index ( https://dhsprogram.com/programming/wealth%20index/Steps_to_c
onstructing_the_new_DHS_Wealth_Index.pdf). However, before start removing the cooking fuel
variable I am make sure that I get the same results, as provided by DHS (
https://dhsprogram.com/topics/wealth-index/Wealth-Index-Cons truction.cfm). I have so far been
through all the DHS countries with available SPSS code but have been unsuccessful in recreating
the wealth index. 

Please see the attachment for specific issues with each survey, however in brief I have the
following issues:

     1.	The domestic, house and land variables do not match the descriptive statistics provided. I
have detail how I have created these based on the guidance. Please could you advice if this is
correct and why I may be getting different results? 

           a.	Domestic has been calculated from V717=6 "Household and domestic" (or 7 "services" if
6 is not available) and the participant is unrelated to household head (v150=12). This is repeated
for the man's questionnaire if available. 

           b.	House: The only question regarding house ownership in most of the countries is in the
individual recode is v745a "Owns a house alone or jointly" with house ownership being defined as
v745a=1 or 2 or 3. Again this is repeated for the Man's questionnaire. 

           c.	Land: The only question regarding working own land in most of the countries is in the
individual recode is v745b "Owns land alone or jointly" with house ownership being defined as
v745b=1 or 2 or 3. Again this is repeated for the Man's questionnaire. 

    2.	Some of the provided results are different to the ones I am getting (see attachments), with
disparities between included variables listed in the SPSS code and those on the excel document.
In addition, some of the results and number of observations are different to those in the excel
spread sheet. Can you please advise why I may be getting these difference and how to solve
these issues? (I happy to provide more details if needed)

    3.	Finally, some of the SPPS code files on the website are corrupted and cannot be read. 

File Attachments
1) Issues with wealth index.xlsx, downloaded 303 times

1) The initial extraction of data for the calculation of the wealth index is done in CSPro using the
raw data file, not the recode file.  Below is the key portion of that code, and following the code a
few notes regarding the domestic, house, and land variables.
PROC HOUSEHOLD
preproc
if AHRESULT <> 1 then
  skip case;
endif;

hhmembers = totocc(AHSEC01_EDT);
hhusual = count(AHSEC01_EDT where AH05 = 1);
hhslept = count(AHSEC01_EDT where AH06 = 1);

domestic = 0;
land = 0;
if AH119 = 1 then land = 1; endif;
house = 0;
{ if ownership of house asked in household questionnaire !!!}
{ if AHXXX = 1 then house = 1; endif; }

postproc
{ Code for exporting the asset variables - used from postproc of household as some variables are
created from the individual questionnaires }
export to wealth_assets case_id (AHCLUST, AHNUMBER)
          AHWEIGHT, AHREGION, AHTYPE, hhmembers, hhusual, hhslept, 
          AHSEC02 exclude(AH108,AH126), 
          AHSEC04 exclude(AH141), 
          domestic, house, land;  { Drop alpha variables !!!}

frequency include(AHREGION, AHTYPE, hhmembers, hhusual, hhslept, 
                  AHSEC02 numeric alpha, 
                  AHSEC04 numeric alpha, domestic, house, land) 
          exclude(AH108,AH126,AH141); { Drop alpha variables !!!}

proc INDIVIDUAL
preproc

{ Woman is a domestic worker and is not related to head of household }
if A913 = maid_code and AH03(ALINE) = not_related then domestic = 1; endif;
{ Woman's partner is a domestic worker, listed in the household and is not related to head of
household }
if A908 = maid_code and A705 in 1:AHMEMBER and AH03(A705) = not_related then domestic =
1; endif;

{ Man is a domestic worker and is not related to head of household }
if AM604 = maid_code and AH03(ALINE) = not_related then domestic = 1; endif;

{ Add code for ownership of land or house if asked in individual women or men's questionnaires -
!!! }
{ Only used to set land or house = 1.  Already initialized to 0 in the preproc for the household }
if AH05(ALINE) = 1 & A925 in 1,2,3 then house = 1; endif;
if AH05(ALINE) = 1 & A928 in 1,2,3 then land  = 1; endif;

if AH05(ALINE) = 1 & AM612 in 1,2,3 then house = 1; endif;
if AH05(ALINE) = 1 & AM615 in 1,2,3 then land  = 1; endif;
Notes:
The default code for maid/domestic (although this varies from survey to survey) is code 51.  See
variables V704, V716, and MV716 for the relevant variables to use. Note also that the person
must not be related to the head of the household (HV101(V003) = 12).  Also when referring to the
husband's occupation in V704, the husband must be listed in the household roster (see V034),
and the husband must not be related to the head of the household (HV101(V034)  12).
For the household and land variables, the respondent to the women's or men's questionnaire
must be de jure members of the household (HV102(V003) = 1/HV102(MV003) = 1).  The house
and land variables may also be set from the household questionnaire.  'Land' is set if HV244 = 1.
'House' may be set if there is a similar variable in the household questionnaire although this is
non-standard (in the standard it is only based on the questions in the man's or the woman's
questionnaires.

2) We will review your notes and check the files available online and get back to you.
3) Can you let us know which files are corrupted, and we will check these.

Thank you very much for your response Trevor, it is greatly appreciated. 

The SPSS text files are: Guinea 2018 (DHS), Nigeria 2018 (DHS), Mali 2018 (DHS)

